

... PPCC PLOT**PURPOSE**

Generates a ppcc plot (that is, a probability plot correlation coefficient plot).

DESCRIPTION

A ppcc plot is a graphical data analysis technique for determining that member of the specified distributional family which provides a "best" distributional fit to the data. The distributional fit will be "best" in the sense that it will have the most linear probability plot of all the selected members of the family. For each of selected members of the distributional family, the probability plot is computed, and the linearity of the probability plot is summarized via the correlation coefficient. The resulting PPCC plot thus consists of:

Vertical axis = probability plot correlation coefficient value;
Horizontal axis = distributional family parameter value.

The value of the distributional parameter (on the horizontal axis) which corresponds to the maximum of the PPCC plot curve (on the vertical axis) is, of course, of interest since it indicates the best-fit member of the family. PPCC plots are available for the following distributional families (with the distributional parameter in parentheses):

1. Tukey lambda (lambda)
2. t (nu)
3. chi-squared (nu)
4. gamma (gamma)
5. extreme value type 2 (gamma)
6. Pareto (gamma)
7. geometric (p)
8. Poisson (lambda)
9. Wald (gamma)
10. inverse Gaussian (gamma)
11. reciprocal inverse gaussian(gamma)
12. fatigue life (gamma)
13. Weibull (gamma)
14. Generalized Pareto (gamma)

SYNTAX 1

<family> PPCC PLOT <x> <SUBSET/EXCEPT/FOR/qualification>

where <x> is the variable of raw data values under analysis;

<family> is one of the following:

TUKEY LAMBDA
T
CHI-SQUARED
GAMMA
EXTREME VALUE TYPE 2
PARETO
GEOMETRIC
POISSON
WALD
INVERSE GAUSSIAN
RIG
FL
WEIBULL
GENERALIZED PARETO
EXTREME VALUE

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for raw data.

SYNTAX 2

<family> PPCC PLOT <y> <x> <SUBSET/EXCEPT/FOR/qualification>

where <y> is the variable of pre-computed frequencies;

<x> is the variable of distinct values for the variable under analysis;

<family> is as above;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

This syntax is used for the case with pre-computed frequencies.

EXAMPLES

LAMBDA PPCC PLOT X

T PPCC PLOT X

EXTREME VALUE TYPE 2 PPCC PLOT X

POISSON PPCC PLOT X

LAMBDA PPCC PLOT F X

T PPCC PLOT F X

EXTREME VALUE TYPE 2 PPCC PLOT F X

POISSON PPCC PLOT F X

NOTE 1

The Weibull, extreme value type II, and generalized Pareto distributions can be based on either the minimum or maximum order statistic. The command SET MINMAX <1/2> is required before the PPCC PLOT command for these distributions. A value of 1 specifies the minimum order statistic and a value of 2 specifies the maximum order statistic. Currently, the generalized Pareto distribution is only supported for the maximum order statistic (i.e., enter SET MINMAX 2).

NOTE 2

The EXTREME VALUE option listed above generates a WEIBULL PPCC PLOT and an EV2 PPCC PLOT overlaid on the same plot.

NOTE 3

The range of parameters is determined automatically. However, if you wish to restrict the range, you can specify the lower and upper limits by appending a 1 or 2 to the parameter name and assigning a value. For example, to restrict a Weibull ppcc plot to values 0.5 and 20, do the following:

```
LET GAMMA1 = 0.5
```

```
LET GAMMA2 = 20
```

```
WEIBULL PPCC PLOT Y
```

NOTE 4

The PPCC PLOT automatically saves several parameters. The MAXPPCC parameter contains the maximum correlation that was computed and the SHAPE parameter contains the value of the estimated distributional parameter (e.g., GAMMA for the Weibull distribution) that corresponds to MAXPPCC.

NOTE 5

The PROBABILITY PLOT command can be used to generate a probability plot for over 35 distributions. For distributions that are actually a family of distributions, the PPCC PLOT is typically employed first to get the best member of the family.

NOTE 6

The PPCC PLOT commands writes conclusions based on the ppcc plot to the DPCONF.TEX (the name may vary depending on the operating system) for some of the distribution families. This file is automatically opened when a DATAPLOT session starts, although only a few commands actually write anything to it.

DEFAULT

None

SYNONYMS

FRECHET and EV2 are synonyms for EXTREME VALUE TYPE 2.

LAMBDA PPCC PLOT and TUKEY PPCC PLOT are synonyms for TUKEY LAMBDA PPCC PLOT.

STUDENT T PPCC PLOT is a synonym for T PPCC PLOT.

The CHISQUARE term can be specified as CHISQUARE or CHI SQUARE.

FL PPCC PLOT, BRIN SAUNDERS PPCC PLOT, and SAUNDERS BRIN PPCC PLOT are synonyms for FATIGUE LIFE PPCC PLOT.

IG PPCC PLOT is a synonym for INVERSE GAUSSIAN PPCC PLOT.

RIG PPCC PLOT is a synonym for RECIPROCAL INVERSE GAUSSIAN PPCC PLOT.

RELATED COMMANDS

FREQUENCY PLOT	=	Generates a frequency plot.
HISTOGRAM	=	Generates a histogram.
PIE CHART	=	Generates a pie chart.
PERCENT POINT PLOT	=	Generates a percent point plot.
PROBABILITY PLOT	=	Generates a probability plot.
PLOT	=	Generates a data or function plot.

REFERENCE

"Continuous Univariate Distributions," 2nd. ed., Johnson, Kotz, and Balkrishnan, John Wiley and Sons, 1994.

"The Probability Plot Correlation Coefficient Test for Normality," James J. Filliben. Technometrics, Vol. 17, No. 1, February 1975.

APPLICATIONS

Distributional Analysis

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```
MULTIPLY 5 3; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE AUTOMATIC
TIC LABEL SIZE 5; LABEL SIZE 5
Y1LABEL PPCC VALUE; X1LABEL PARAMETER VALUE
X1LABEL DISPLACEMENT 8
HEIGHT 5
JUSTIFICATION CENTER
SET MINMAX 1
.
LET LAMBDA = 1.5
LET Y = TUKEY LAMBDA RANDOM NUMBERS FOR I = 1 1 100
TUKEY LAMBDA PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET LAMBDA = 25
LET Y = POISSON RANDOM NUMBERS FOR I = 1 1 100
POISSON PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET NU = 25
LET Y = T RANDOM NUMBERS FOR I = 1 1 100
T PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET NU = 5
LET Y = CHI-SQUARE RANDOM NUMBERS FOR I = 1 1 100
CHI-SQUARE PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET P = 0.5
LET Y = GEOMETRIC RANDOM NUMBERS FOR I = 1 1 100
GEOMETRIC PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET GAMMA = 0.5
LET Y = GAMMA RANDOM NUMBERS FOR I = 1 1 100
```

```

GAMMA PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = EV2 RANDOM NUMBERS FOR I = 1 1 100
EV2 PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = PARETO RANDOM NUMBERS FOR I = 1 1 100
PARETO PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = WALD RANDOM NUMBERS FOR I = 1 1 100
WALD PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = INVERSE GAUSSIAN RANDOM NUMBERS FOR I = 1 1 100
INVERSE GAUSSIAN PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = RECIPRICAL INVERSE GAUSSIAN RANDOM NUMBERS FOR I = 1 1 100
RECIPRICAL INVERSE GAUSSIAN PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = FATIGUE LIFE RANDOM NUMBERS FOR I = 1 1 100
FATIGUE LIFE PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
LET Y = WEIBULL RANDOM NUMBERS FOR I = 1 1 100
WEIBULL PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
EXTREME VALUE PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
SET MINMAX 2
LET Y = GENERALIZED PARETO RANDOM NUMBERS FOR I = 1 1 100
GENERALIZED PARETO PPCC PLOT Y
MOVE 50 2; TEXT SHAPE = ^SHAPE
END OF MULTILOT
    
```

